NASA TECHNICAL NOTE

NASA TN D-6871

c.1

NASA TN D-6871

TECH LIBRARY KAFB, NM

0133726

LOAN COPY: RETURN
AFWL (DOUL)
KIRTLAND AFB, N. M

NUMERICAL CALCULATION OF BOUNDARY-INDUCED INTERFERENCE IN SLOTTED OR PERFORATED WIND TUNNELS INCLUDING VISCOUS EFFECTS IN SLOTS

by James D. Keller
Langley Research Center
Hampton, Va. 23365

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • AUGUST 1972

TECH LIBRARY KAFB, NM

0133726

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.
NASA TN D-6871		
4. Title and Subtitle NUMERICAL CALCULATION OF BOUNDARY-INDUCED INTERFERENCE IN SLOTTED OR PERFORATED WIND TUNNELS INCLUDING VISCOUS EFFECTS IN SLOTS		5. Report Date August 1972 6. Performing Organization Code
7. Author(s) James D. Keller		8. Performing Organization Report No. L-8349
9. Performing Organization Name and Address NASA Langley Research Center Hampton, Va. 23365		10. Work Unit No. 501-06-09-02 11. Contract or Grant No.
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, D.C. 20546		13. Type of Report and Period Covered Technical Note 14. Sponsoring Agency Code
15. Supplementary Notes		

16. Abstract

A numerical method is presented for calculating the incompressible boundary-induced interference in wind tunnels of rectangular cross section with slotted or perforated walls. The method includes a wall representation which is capable of satisfying a generalized homogeneous boundary condition including the effects of viscosity within the slots. The effects of viscosity in the slots are found to be very significant. The method allows for a variation in the boundary conditions along the tunnel walls. The model can be any configuration and can be located anywhere in the test section. The interference can be computed at any point in the test section.

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Wind-tunnel interference Perforated wind tunnels Slotted wind tunnels	Unclassified – Unlimited

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price*
Unclassified	Unclassified	22	$3.00

*For sale by the National Technical Information Service, Springfield, Virginia 22151

NUMERICAL CALCULATION OF BOUNDARY-INDUCED INTERFERENCE IN SLOTTED OR PERFORATED WIND TUNNELS INCLUDING VISCOUS EFFECTS IN SLOTS

By James D. Keller
Langley Research Center

SUMMARY

A numerical method is presented for calculating the incompressible boundary-induced interference in wind tunnels of rectangular cross section with slotted or perforated walls. The method includes a wall representation which is capable of satisfying a generalized homogeneous boundary condition including the effects of viscosity within the slots. The effects of viscosity in the slots are found to be very significant. The method allows for a variation in the boundary conditions along the tunnel walls. The model can be any configuration and can be located anywhere in the test section. The interference can be computed at any point in the test section.

INTRODUCTION

In order to obtain accurate wind-tunnel data, the measured quantities must often be corrected to account for the interference caused by the wind-tunnel boundaries. Theoretical methods are presently available for predicting the interference due to the tunnel walls in certain cases. The analytical methods are limited to infinite-length test sections with constant wall properties in the tunnel stream direction. Some methods are limited as to model size, position, and load distribution. A numerical method for calculating the boundary-induced interference in ventilated wind tunnels is presented in reference 1. The method consists of dividing the tunnel walls into rectangular elements which are each represented by a source distribution. A matrix equation is then solved to find the source strengths which allow the boundary conditions to be satisfied at the centroid of each element. In reference 1 each element was represented by a source distribution of constant strength over the element. This representation is particularly well suited to satisfying an ideal slotted-wall boundary condition. The ideal slotted-wall boundary condition, however, is only a special case of a more general boundary condition which can include the effects of viscosity within the slots.

The present investigation deals specifically with a modified representation for the tunnel walls which is suitable for the satisfaction of the more general boundary condition including the effects of viscosity in the slots. The method presented is limited to incompressible flow and cannot handle the usual assumption of a test section which extends to infinity upstream and downstream of the model. The method also requires the experimental determination of one of the parameters in the boundary condition. The method has broad applicability, however, because the boundary conditions on the tunnel walls may vary almost without limit. The model representation is also quite general. The model may be located anywhere in the test section and at any orientation. A sample computer program used in making the calculations is given in an appendix.

SYMBOLS

a	effect of one element on another
b	effect of model on an element
c_1,c_2,c_3,c_4	coefficients in equation (4)
d	distance between slot centers
l	slot parameter
N	total number of elements
n	direction normal to wall
R	restriction parameter
s	wing span
t	slot width
w_w	upwash velocity caused by tunnel walls
x,y,z	Cartesian coordinates
Γ_m	circulation of model
δ	lift interference factor
ξ,η,ζ	Cartesian coordinates

σ source distribution strength

$\sigma' = \frac{d\sigma}{dx}$

φ perturbation velocity potential function

φ^* velocity potential function for an element divided by σ' for the element

Subscripts:

i at ith element

j at jth element

L downstream end of each source distribution

m due to model

w due to tunnel walls

ANALYSIS

General Statement of Problem

The governing equation used in the analysis of incompressible wind-tunnel interference is

$$\frac{\partial^2 \varphi}{\partial x^2} + \frac{\partial^2 \varphi}{\partial y^2} + \frac{\partial^2 \varphi}{\partial z^2} = 0 \tag{1}$$

where φ is the perturbation velocity potential function for the entire flow field. Let $\varphi = \varphi_m + \varphi_w$ where φ_m is the potential function of the disturbances due to the model in free air and φ_w is the potential function of the additional flow due to the tunnel walls. If φ_m is taken as a known solution of equation (1) which approximates the flow field at a distance from the model in free air, then φ_w can be determined by the fact that φ must satisfy certain boundary conditions at the tunnel walls. The objective in determining φ_w is to be able to calculate the change in the free-stream conditions caused by the tunnel walls. Since φ_m needs to be known only on the tunnel walls, any inaccuracies in φ_m near the model will have no effect on the determination of φ_w.

Boundary Conditions

The boundary condition to be satisfied at a solid boundary is that there can be no flow through the boundary; that is

$$\frac{\partial\varphi}{\partial n} = 0$$

where n is the direction normal to the wall (positive outwards). The boundary condition to be satisfied at an open jet boundary is that there be no pressure difference across the boundary. This boundary condition can be approximated by (ref. 2)

$$\frac{\partial\varphi}{\partial x} = 0$$

Reference 3 gives the homogeneous boundary condition to be satisfied at a perforated wall as

$$\frac{\partial\varphi}{\partial x} + \frac{1}{R}\frac{\partial\varphi}{\partial n} = 0$$

where R is a restriction parameter which relates the pressure difference across the wall to the flow through the wall. In practice, R must be determined experimentally for a given wall. A detailed discussion of the restriction parameter for both porous and perforated walls can be found in reference 4.

The homogeneous boundary condition to be satisfied at a wall with several longitudinal slots is given in reference 5 as

$$\varphi + l\frac{\partial\varphi}{\partial n} = 0 \tag{2}$$

where l is a slot parameter given by

$$l = \frac{d}{\pi}\ln\csc\left(\frac{\pi}{2}\frac{t}{d}\right)$$

where t is the slot width and d is the distance between slot centers. This slot parameter was derived on the basis of two-dimensional flow, and it is assumed that it can be applied at each location in the tunnel even if the slot width varies. Equation (2) can be differentiated with respect to x to give

$$\frac{\partial\varphi}{\partial x} + \frac{\partial}{\partial x}\left(l\frac{\partial\varphi}{\partial n}\right) = 0 \tag{3}$$

For constant slot width, equation (3) becomes

$$\frac{\partial\varphi}{\partial x} + l\frac{\partial^2\varphi}{\partial x\,\partial n} = 0$$

which is the form given by many authors.

The ideal slotted-wall boundary condition was derived on the basis of inviscid flow at the slots. In reference 6 the addition of another term to account for the effects of viscosity in the slots is suggested. The boundary condition is then

$$\frac{\partial\varphi}{\partial x} + \frac{1}{R}\frac{\partial\varphi}{\partial n} + l\,\frac{\partial^2\varphi}{\partial x\,\partial n} = 0$$

If the coefficient $1/R$ is replaced by $\frac{1}{R} + \frac{\partial l}{\partial x}$, this boundary condition will also apply to walls which do not have constant slot width. As R approaches infinity, this boundary condition approaches that for an ideal slotted wall. However, as pointed out in reference 7, the ideal condition is not always valid. Experiments have shown that for typical test-section configurations, R can be of the order of unity (ref. 8). This value of R will have a very pronounced effect on the interference in the test section. Thus, it is important to consider the effects of viscosity in the slots and retain the additional term in the boundary condition. Note that no theoretical method exists for determining the value of R for a particular test section. It must be determined experimentally. This experimental determination of the restriction parameter might be quite difficult because for a tunnel with varying slot width the restriction parameter may vary with position also.

In this paper, a general boundary condition of the form

$$c_1\varphi + c_2\frac{\partial\varphi}{\partial x} + c_3\frac{\partial\varphi}{\partial n} + c_4\frac{\partial^2\varphi}{\partial x\,\partial n} = 0 \tag{4}$$

is considered. This boundary condition contains all previous conditions as special cases as shown in the following table:

Type of boundary condition	c_1	c_2	c_3	c_4
Closed wall	0	0	1	0
Open jet	0	1	0	0
Perforated wall	0	1	$\frac{1}{R}$	0
Ideal slotted wall (integrated form)	1	0	l	0
Ideal slotted wall (differentiated form)	0	1	$\frac{\partial l}{\partial x}$	l
Slotted wall including viscosity in slots	0	1	$\frac{\partial l}{\partial x} + \frac{1}{R}$	l

Additional discussions of these boundary conditions may be found in references 4, 6, and 7.

Representation of Tunnel Walls

In order to satisfy the homogeneous boundary condition, the tunnel walls are divided into longitudinal strips and each strip is divided into a number of rectangular elements. The boundary condition will be satisfied at the centroid of each element. The coordinate system to be used has the X-axis extending along the tunnel center line, with the positive direction being the tunnel stream direction. The Z-axis is positive upwards, and the Y-axis is chosen so that the coordinate system is a right-handed system. In reference 1 each tunnel-wall element was represented by a source distribution of constant strength over the element. This representation is particularly suited to the satisfaction of the ideal slotted-wall boundary condition in integrated form (eq. (2)) because for this case the matrix of influence coefficients is diagonally dominant. This diagonal dominance also holds for the perforated-wall boundary condition for small values of R. However, for the ideal slotted-wall boundary condition in differentiated form or for large values of R in the general slotted-wall and perforated-wall boundary conditions, this representation would lead to elements on the diagonal of the matrix of influence coefficients which are either zero or very small. This nearly singular matrix leads to numerical difficulties and inaccuracies in trying to solve the resulting system of equations. In order to avoid these difficulties, let each tunnel-wall element be represented by a source distribution over the element and downstream of the element at least to the end of the strip. The strength of the source distribution σ varies linearly, with a slope $\sigma' = \frac{d\sigma}{dx}$, on the element itself and then remains constant downstream of the element. This representation is used so as to have the source strength continuous along a strip and still have only one unknown (the source strength slope) for each element. Thus, the source strength is zero at the upstream end of the strip and varies in linear segments along the length of the strip. If φ^* is the potential function for a particular element divided by the source strength slope σ' for that element, then

$$\varphi_W = \sum_{j=1}^{N} \varphi_j^* \sigma_j' \tag{5}$$

where N is the total number of elements.

Consider an element in the top or bottom wall with corners as shown in figure 1.



Figure 1.- Schematic of an element in top or bottom wall.

The potential function at a point (x,y,z) due to this source distribution is

$$\varphi^* = -\int_{\xi_1}^{\xi_2}\int_{\eta_1}^{\eta_2} \frac{(\xi - \xi_1)d\eta\, d\xi}{\sqrt{(x-\xi)^2 + (y-\eta)^2 + (z-\zeta_1)^2}} - \int_{\xi_2}^{\xi_L}\int_{\eta_1}^{\eta_2} \frac{(\xi_2 - \xi_1)d\eta\, d\xi}{\sqrt{(x-\xi)^2 + (y-\eta)^2 + (z-\zeta_1)^2}}$$

This potential function and its derivatives must be evaluated to satisfy the boundary conditions. For convenience in writing the equations, let $x_1 = x - \xi_1$, $x_2 = x - \xi_2$, $x_L = x - \xi_L$, $y_1 = y - \eta_1$, $y_2 = y - \eta_2$, and $z_1 = z - \zeta_1$. The required equations are then

$$\begin{aligned}
\varphi^* = {} & \frac{y_2}{2}\sqrt{x_2^2 + y_2^2 + z_1^2} - \frac{y_1}{2}\sqrt{x_2^2 + y_1^2 + z_1^2} - \frac{y_2}{2}\sqrt{x_1^2 + y_2^2 + z_1^2} + \frac{y_1}{2}\sqrt{x_1^2 + y_1^2 + z_1^2} + \frac{z_1^2 - x_2^2}{2}\ln\left(\frac{y_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}{y_1 + \sqrt{x_2^2 + y_1^2 + z_1^2}}\right) \\
& - \frac{z_1^2 - x_1^2}{2}\ln\left(\frac{y_2 + \sqrt{x_1^2 + y_2^2 + z_1^2}}{y_1 + \sqrt{x_1^2 + y_1^2 + z_1^2}}\right) - x_1 y_2 \ln\left(\frac{x_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}{x_1 + \sqrt{x_1^2 + y_2^2 + z_1^2}}\right) + x_1 y_1 \ln\left(\frac{x_2 + \sqrt{x_2^2 + y_1^2 + z_1^2}}{x_1 + \sqrt{x_1^2 + y_1^2 + z_1^2}}\right) \\
& - (\xi_2 - \xi_1)y_2 \ln\left(\frac{x_L + \sqrt{x_L^2 + y_2^2 + z_1^2}}{x_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}\right) + (\xi_2 - \xi_1)y_1 \ln\left(\frac{x_L + \sqrt{x_L^2 + y_1^2 + z_1^2}}{x_2 + \sqrt{x_2^2 + y_1^2 + z_1^2}}\right) - (\xi_2 - \xi_1)x_L \ln\left(\frac{y_2 + \sqrt{x_L^2 + y_2^2 + z_1^2}}{y_1 + \sqrt{x_L^2 + y_1^2 + z_1^2}}\right) \\
& + x_1|z_1|\left\{\tan^{-1}\left[\frac{x_2 y_2}{\sqrt{z_1^2(x_2^2 + y_2^2 + z_1^2)}}\right] - \tan^{-1}\left[\frac{x_1 y_2}{\sqrt{z_1^2(x_1^2 + y_2^2 + z_1^2)}}\right] - \tan^{-1}\left[\frac{x_2 y_1}{\sqrt{z_1^2(x_2^2 + y_1^2 + z_1^2)}}\right] + \tan^{-1}\left[\frac{x_1 y_1}{\sqrt{z_1^2(x_1^2 + y_1^2 + z_1^2)}}\right]\right\} \\
& + (\xi_2 - \xi_1)|z_1|\left\{\tan^{-1}\left[\frac{x_L y_2}{\sqrt{z_1^2(x_L^2 + y_2^2 + z_1^2)}}\right] - \tan^{-1}\left[\frac{x_2 y_2}{\sqrt{z_1^2(x_2^2 + y_2^2 + z_1^2)}}\right] - \tan^{-1}\left[\frac{x_L y_1}{\sqrt{z_1^2(x_L^2 + y_1^2 + z_1^2)}}\right] + \tan^{-1}\left[\frac{x_2 y_1}{\sqrt{z_1^2(x_2^2 + y_1^2 + z_1^2)}}\right]\right\}
\end{aligned} \tag{6}$$

$$\begin{aligned}
\frac{\partial\varphi^*}{\partial x} = {} & y_2 \ln\left(\frac{x_1 + \sqrt{x_1^2 + y_2^2 + z_1^2}}{x_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}\right) + y_1 \ln\left(\frac{x_2 + \sqrt{x_2^2 + y_1^2 + z_1^2}}{x_1 + \sqrt{x_1^2 + y_1^2 + z_1^2}}\right) + x_1 \ln\left[\left(\frac{y_1 + \sqrt{x_2^2 + y_1^2 + z_1^2}}{y_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}\right)\left(\frac{y_2 + \sqrt{x_1^2 + y_2^2 + z_1^2}}{y_1 + \sqrt{x_1^2 + y_1^2 + z_1^2}}\right)\right] \\
& + (\xi_2 - \xi_1)\ln\left[\left(\frac{y_1 + \sqrt{x_L^2 + y_1^2 + z_1^2}}{y_2 + \sqrt{x_L^2 + y_2^2 + z_1^2}}\right)\left(\frac{y_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}{y_1 + \sqrt{x_2^2 + y_1^2 + z_1^2}}\right)\right] + |z_1|\left\{\tan^{-1}\left[\frac{x_2 y_2}{\sqrt{z_1^2(x_2^2 + y_2^2 + z_1^2)}}\right]\right. \\
& \left. - \tan^{-1}\left[\frac{x_1 y_2}{\sqrt{z_1^2(x_1^2 + y_2^2 + z_1^2)}}\right] - \tan^{-1}\left[\frac{x_2 y_1}{\sqrt{z_1^2(x_2^2 + y_1^2 + z_1^2)}}\right] + \tan^{-1}\left[\frac{x_1 y_1}{\sqrt{z_1^2(x_1^2 + y_1^2 + z_1^2)}}\right]\right\}
\end{aligned} \tag{7}$$

$$\frac{\partial\varphi^*}{\partial y} = \sqrt{x_2^2 + y_2^2 + z_1^2} - \sqrt{x_1^2 + y_2^2 + z_1^2} - \sqrt{x_2^2 + y_1^2 + z_1^2} + \sqrt{x_1^2 + y_1^2 + z_1^2}$$

$$+ x_1 \ln\left[\left(\frac{x_1 + \sqrt{x_1^2 + y_2^2 + z_1^2}}{x_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}\right)\left(\frac{x_2 + \sqrt{x_2^2 + y_1^2 + z_1^2}}{x_1 + \sqrt{x_1^2 + y_1^2 + z_1^2}}\right)\right] + (\xi_2 - \xi_1) \ln\left[\left(\frac{x_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}{x_L + \sqrt{x_L^2 + y_2^2 + z_1^2}}\right)\left(\frac{x_L + \sqrt{x_L^2 + y_1^2 + z_1^2}}{x_2 + \sqrt{x_2^2 + y_1^2 + z_1^2}}\right)\right] \tag{8}$$

$$\frac{\partial^2\varphi^*}{\partial x\,\partial y} = \ln\left[\left(\frac{x_1 + \sqrt{x_1^2 + y_2^2 + z_1^2}}{x_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}\right)\left(\frac{x_2 + \sqrt{x_2^2 + y_1^2 + z_1^2}}{x_1 + \sqrt{x_1^2 + y_1^2 + z_1^2}}\right)\right] + \frac{\xi_2 - \xi_1}{\sqrt{x_L^2 + y_1^2 + z_1^2}} - \frac{\xi_2 - \xi_1}{\sqrt{x_L^2 + y_2^2 + z_1^2}} \tag{9}$$

$$\frac{\partial\varphi^*}{\partial z} = z_1 \ln\left[\left(\frac{y_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}{y_1 + \sqrt{x_2^2 + y_1^2 + z_1^2}}\right)\left(\frac{y_1 + \sqrt{x_1^2 + y_1^2 + z_1^2}}{y_2 + \sqrt{x_1^2 + y_2^2 + z_1^2}}\right)\right]$$

$$+ x_1 \operatorname{sgn}(z_1)\left\{\tan^{-1}\left[\frac{x_2 y_2}{\sqrt{z_1^2\left(x_2^2 + y_2^2 + z_1^2\right)}}\right] - \tan^{-1}\left[\frac{x_1 y_2}{\sqrt{z_1^2\left(x_1^2 + y_2^2 + z_1^2\right)}}\right] - \tan^{-1}\left[\frac{x_2 y_1}{\sqrt{z_1^2\left(x_2^2 + y_1^2 + z_1^2\right)}}\right] + \tan^{-1}\left[\frac{x_1 y_1}{\sqrt{z_1^2\left(x_1^2 + y_1^2 + z_1^2\right)}}\right]\right\}$$

$$+ (\xi_2 - \xi_1)\operatorname{sgn}(z_1)\left\{\tan^{-1}\left[\frac{x_L y_2}{\sqrt{z_1^2\left(x_L^2 + y_2^2 + z_1^2\right)}}\right] - \tan^{-1}\left[\frac{x_2 y_2}{\sqrt{z_1^2\left(x_2^2 + y_2^2 + z_1^2\right)}}\right] - \tan^{-1}\left[\frac{x_L y_1}{\sqrt{z_1^2\left(x_L^2 + y_1^2 + z_1^2\right)}}\right] + \tan^{-1}\left[\frac{x_2 y_1}{\sqrt{z_1^2\left(x_2^2 + y_1^2 + z_1^2\right)}}\right]\right\} \tag{10}$$

and

$$\frac{\partial^2\varphi^*}{\partial x\,\partial z} = \operatorname{sgn}(z_1)\left\{\tan^{-1}\left[\frac{x_2 y_2}{\sqrt{z_1^2\left(x_2^2 + y_2^2 + z_1^2\right)}}\right] - \tan^{-1}\left[\frac{x_1 y_2}{\sqrt{z_1^2\left(x_1^2 + y_2^2 + z_1^2\right)}}\right] - \tan^{-1}\left[\frac{x_2 y_1}{\sqrt{z_1^2\left(x_2^2 + y_1^2 + z_1^2\right)}}\right]\right.$$

$$\left. + \tan^{-1}\left[\frac{x_1 y_1}{\sqrt{z_1^2\left(x_1^2 + y_1^2 + z_1^2\right)}}\right]\right\} + \frac{(\xi_2 - \xi_1) z_1}{x_L^2 + z_1^2}\left(\frac{y_2}{\sqrt{x_L^2 + y_2^2 + z_1^2}} - \frac{y_1}{\sqrt{x_L^2 + y_1^2 + z_1^2}}\right) \tag{11}$$

In order to find the effect of an element in a side wall, y and z must be interchanged in equations (6) to (11). By examining equations (10) and (11) for $z_1 \to 0$, the effect of an element at its own centroid is found to be

$$\frac{\partial \varphi^*}{\partial n} = -2\pi \frac{\xi_2 - \xi_1}{2}$$

and

$$\frac{\partial^2 \varphi^*}{\partial x \, \partial n} = -2\pi$$

At points of the same strip but downstream of the element

$$\frac{\partial \varphi^*}{\partial n} = -2\pi(\xi_2 - \xi_1)$$

and

$$\frac{\partial^2 \varphi^*}{\partial x \, \partial n} = 0$$

At all other elements of the wall in which an element is located, its contributions to $\partial\varphi^*/\partial n$ and $\partial^2\varphi^*/\partial x \, \partial n$ are zero.

Computation of Source Strength Slopes

In order to compute the source strength slopes required to satisfy the boundary conditions at the centroid of each element, a matrix equation is needed to express these boundary conditions. Let a_{ij} be the effect at the centroid of the ith element due to the source distribution corresponding to the jth element $\left(a = c_1\varphi^* + c_2 \frac{\partial \varphi^*}{\partial x} + c_3 \frac{\partial \varphi^*}{\partial n} + c_4 \frac{\partial^2 \varphi^*}{\partial x \, \partial n}\right)$. Let b_i be the effect of the model at the centroid of the ith element $\left(b = c_1\varphi_m + c_2 \frac{\partial \varphi_m}{\partial x} + c_3 \frac{\partial \varphi_m}{\partial n} + c_4 \frac{\partial^2 \varphi_m}{\partial x \, \partial n}\right)$. Then the matrix equation which expresses the boundary condition is

$$A\Sigma' = -B \tag{12}$$

where

$$A = \left[a_{ij}\right]$$

$$\Sigma' = \left[\sigma_j'\right]$$

and

$$B = \left[b_i\right]$$

Equation (12) can be solved for the values of σ_j' which can then be used to compute the interference potential due to the tunnel walls through the use of equation (5).

RESULTS AND DISCUSSION

As a relatively simple example, consider the lift interference due to a small lifting wing mounted in the center of a square test section with solid side walls and four equally spaced slots in the top and bottom walls. Each tunnel wall is divided into four strips of equal width and each strip is divided into 10 elements by cutting planes at $x = -1.00$, -0.70, -0.45, -0.25, -0.10, 0.00, 0.10, 0.25, 0.45, 0.70, and 1.00. (For convenience, the tunnel width is taken to be unity.)

The wing is represented by a horseshoe vortex of circulation Γ_m. The span s of the horseshoe vortex is assumed to be so small that it becomes a vortex doublet starting at (0,0,0). The perturbation velocity potential function φ_m at a point (x,y,z) due to this representation of the model is given by

$$\varphi_m = \frac{\Gamma_m s}{4\pi} \frac{z}{y^2 + z^2}\left(1 + \frac{x}{\sqrt{x^2 + y^2 + z^2}}\right)$$

so that

$$\frac{1}{\Gamma_m s} \frac{\partial \varphi_m}{\partial x} = \frac{1}{4\pi} \frac{z}{\left(x^2 + y^2 + z^2\right)^{3/2}}$$

$$\frac{1}{\Gamma_m s} \frac{\partial \varphi_m}{\partial y} = \frac{-1}{4\pi\left(y^2 + z^2\right)^2}\left[2yz + \frac{2x^3yz + 3xy^3z + 3xyz^3}{\left(x^2 + y^2 + z^2\right)^{3/2}}\right]$$

$$\frac{1}{\Gamma_m s} \frac{\partial^2 \varphi_m}{\partial x\, \partial y} = -\frac{3}{4\pi} \frac{yz}{\left(x^2 + y^2 + z^2\right)^{5/2}}$$

$$\frac{1}{\Gamma_m s} \frac{\partial \varphi_m}{\partial z} = \frac{1}{4\pi\left(y^2 + z^2\right)^2}\left[y^2 - z^2 + \frac{x^3y^2 + xy^4 - x^3z^2 - xy^2z^2 - 2xz^4}{\left(x^2 + y^2 + z^2\right)^{3/2}}\right]$$

and

$$\frac{1}{\Gamma_m s} \frac{\partial^2 \varphi_m}{\partial x\, \partial z} = \frac{1}{4\pi} \frac{x^2 + y^2 - 2z^2}{\left(x^2 + y^2 + z^2\right)^{5/2}}$$

These quantities are used on the right-hand side of equation (12) which is then solved for $\sigma_j'/\Gamma_m s$. The values of $\sigma_j'/\Gamma_m s$ are suitable for the computation of the upwash velocity

$\frac{w_w}{\Gamma_m s} = \frac{1}{\Gamma_m s} \frac{\partial \varphi_w}{\partial z}$ at any point in the test section by summing the velocity due to each element. The lift interference factor (ref. 5) is then

$$\delta = \frac{1}{2} \frac{w_w}{\Gamma_m s}$$

Figure 2 shows the lift interference factor at the center of the tunnel as a function of the ratio of the slot width to the distance between slot centers. The results were computed by using the ideal slotted-wall boundary condition in both integrated and differentiated forms. The results computed by using the integrated form of the boundary condition



Figure 2.- Lift interference factor at vanishingly small-span wing in square tunnel with four slots in top and bottom walls.

(solid-line curve) are the same as those computed by using the wall representation of reference 1. The difference in the lift interference factor when the boundary condition is used in the two different forms is due to the fact that the differentiated form of the boundary condition does not satisfy the additional requirement that there be no disturbance due to the tunnel walls at an infinite distance upstream of the model. However, the addition of elements to the upstream end of the test section rapidly eliminates the difference. When the test section starts three tunnel widths upstream of the model, the difference is very nearly zero. The reason for this can be seen more clearly in figure 3 which shows



Figure 3.- Variation of source strength along wall.

a typical variation of the source distribution strength along one of the strips of the top wall. It can be seen that the distribution which corresponds to the differentiated form of the boundary condition would have the same value at the centroid of each element as the other distributions if the source strength was increased by a constant amount. This constant shift corresponds to the source strength which would have been built up by the point $x = -1$ on an infinitely long test section. The error caused by using the differentiated form of the boundary condition comes about not so much because the source distribution on the far upstream portion of the test section is not included, but rather because the source strength which would have been built up on the far upstream portion of an infinitely long test section is not included all along the rest of the test section. Putting additional elements farther upstream of the model almost completely eliminates this error, but requires a larger matrix which takes up more computer storage and time. On the downstream end, the source distribution which extends downstream from each element can be extended beyond the end of the last element if desired.

Figure 4 shows the lift interference factor at the center of the tunnel with the general slotted-wall boundary condition used for several values of the restriction parameter R. This figure shows that the effects of viscosity in the slots can be very significant.



Figure 4.- Lift interference factor at vanishingly small-span wing in square tunnel with four slots in top and bottom walls for several values of R.

The results presented here are for the simple case of a small-span wing mounted in the center of a tunnel with constant slot width and constant restriction parameter. The method, however, is applicable to more general problems. The model representation can be quite general, including the case of a large-span swept wing with nonuniform loading located anywhere in the test section. The slot width and restriction parameter may vary with position on the boundary. The results presented here are also for the case of a test section which extends one tunnel width upstream and downstream of the model. This arrangement of tunnel-wall elements was used for several reasons. First, the computer time required to invert a matrix of this size (N = 160) is not too large (about 1 minute on a Control Data 6600 computer system). Second, the results can be compared directly with those of reference 1 which used the same arrangement of elements but a different source distribution to represent each element. Third, use of this tunnel length clearly shows the error caused by using the differentiated form of the ideal slotted-wall boundary condition and not including the portion of the boundary far upstream of the model. Greater accuracy could be obtained by adding elements to the test section, particularly farther upstream of the model, and by letting the constant strength source distribution which trails downstream of each element to extend beyond the end of the last element to a large distance downstream of the model. In equations (7) to (11), x_L can be allowed to approach minus infinity and simplify the equations somewhat. This is not possible in equation (6), so when the ideal slotted-wall boundary condition is used in integrated form, the test section must be terminated at a finite distance downstream of the model. Although the present development is oriented toward tunnels of rectangular cross section, it could be extended to other cross-sectional shapes.

The sample computer program given in the appendix is not the one used to compute the results shown here. It has additional elements farther upstream and downstream of the model for greater accuracy. In the sample program the elements extend from $x = -2.6$ to $x = 1.8$, and the constant strength portion downstream of each element extends to $\xi_L = 10$. This program requires about $240\ 000_8$ storage locations and about 5 minutes on a Control Data 6600 computer system.

CONCLUDING REMARKS

A numerical method for calculating the boundary-induced interference in slotted or perforated wind tunnels has been presented. The method includes a wall representation which is capable of satisfying a generalized boundary condition including the effects of viscosity within the slots of a slotted-wall tunnel. The method is limited to incompressible flow and requires the experimental determination of one of the coefficients in the boundary condition. It is also limited to finite-length test sections.

The method presented has broad applicability and allows for the boundary condition to vary over the test-section walls. This feature should aid in the design of test sections which have nearly constant interference over the space occupied by the model. The model representation is also quite general, including the case of a large-span swept wing with nonuniform loading located anywhere in the test section. The interference can also be computed anywhere in the test section. The effects of viscosity in the slots are found to be very significant.

Langley Research Center,
National Aeronautics and Space Administration,
Hampton, Va., June 30, 1972.

APPENDIX

SAMPLE FORTRAN PROGRAM

THIS APPENDIX CONTAINS A SAMPLE FORTRAN PROGRAM FOR COMPUTING THE LIFT INTERFERENCE FACTOR IN A WIND TUNNEL OF RECTANGULAR CROSS SECTION WITH SLOTTED OR PERFORATED WALLS. THE PROGRAM WAS WRITTEN FOR USE ON CDC 6000 SERIES COMPUTERS. IT IS INTENDED ONLY AS A SAMPLE. MODIFICATIONS MUST BE MADE TO THE PROGRAM IN ORDER TO COMPUTE DIFFERENT CASES. IF IT IS DESIRED TO REDUCE THE MATRIX SIZE, THE PROGRAM GIVEN IN REFERENCE 1 COULD BE MODIFIED USING THE ARITHMETIC STATEMENT FUNCTIONS GIVEN HERE. FORMAL INPUTS CAN BE FOUND ON LINES 208 AND 269.

```
      PROGRAM A3771(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
2     DIMENSION XI(256),ETA(256),ZETA(256),XI1(256),XI2(256),ETA1(256)
3    * ,ETA2(256),ZETA1(256),ZETA2(256),A(256,256),B(256),SIGMA(256)
4    * ,C1(256),C2(256),C3(256),C4(256)
5     DIMENSION XA(10),YA(10),WT(10)
6     SGN(X)=SIGN(1.0,X)
7     P(X1,X2,Y1,Y2,Z)=Y2/2.*SQRT(X2*X2+Y2*Y2+Z*Z)-Y1/2.*SQRT(X2*X2+Y1*
8    * Y1+Z*Z)-Y2/2.*SQRT(X1*X1+Y2*Y2+Z*Z)+Y1/2.*SQRT(X1*X1+Y1*Y1+Z*Z)
9    * +(Z*Z-X2*X2)/2.*ALOG(ABS((Y2+SQRT(X2*X2+Y2*Y2+Z*Z))/(Y1+SQRT(X2*
10   * X2+Y1*Y1+Z*Z)))))-(Z*Z-X1*X1)/2.*ALOG(ABS((Y2+SQRT(X1*X1+Y2*Y2+Z*Z
11   * ))/(Y1+SQRT(X1*X1+Y1*Y1+Z*Z))))-X1*Y2*ALOG(ABS((X2+SQRT(X2*X2+Y2*
12   * Y2+Z*Z))/(X1+SQRT(X1*X1+Y2*Y2+Z*Z))))+X1*Y1*ALOG(ABS((X2+SQRT(X2*
13   * X2+Y1*Y1+Z*Z))/(X1+SQRT(X1*X1+Y1*Y1+Z*Z))))-XD*X1*ALOG(ABS((Y2+
14   * SQRT(X1*X1+Y2*Y2+Z*Z))/(Y1+SQRT(X1*X1+Y1*Y1+Z*Z))))-XD*Y2*ALOG(
15   * ABS((X1+SQRT(X1*X1+Y2*Y2+Z*Z))/(X2+SQRT(X2*X2+Y2*Y2+Z*Z))))+XD*Y1
16   * *ALOG(ABS((X1+SQRT(X1*X1+Y1*Y1+Z*Z))/(X2+SQRT(X2*X2+Y1*Y1+Z*Z))))
17   * -X1*ABS(Z)*(-ATAN(X2*Y2/ABS(Z)/SQRT(X2*X2+Y2*Y2+Z*Z))+ATAN(X1*Y2/
18   * ABS(Z)/SQRT(X1*X1+Y2*Y2+Z*Z))+ATAN(X2*Y1/ABS(Z)/SQRT(X2*X2+Y1*Y1+
19   * Z*Z))-ATAN(X1*Y1/ABS(Z)/SQRT(X1*X1+Y1*Y1+Z*Z)))-XD*ABS(Z)*(-ATAN(
20   * X1*Y2/ABS(Z)/SQRT(X1*X1+Y2*Y2+Z*Z))+ATAN(X2*Y2/ABS(Z)/SQRT(X2*X2+
21   * Y2*Y2+Z*Z))+ATAN(X1*Y1/ABS(Z)/SQRT(X1*X1+Y1*Y1+Z*Z))-ATAN(X2*Y1/
22   * ABS(Z)/SQRT(X2*X2+Y1*Y1+Z*Z)))
23    S(X1,X2,Y1,Y2,Z)=Y2/2.*SQRT(X2*X2+Y2*Y2+Z*Z)-Y1/2.*SQRT(X2*X2+Y1*
24   * Y1+Z*Z)-Y2/2.*SQRT(X1*X1+Y2*Y2+Z*Z)+Y1/2.*SQRT(X1*X1+Y1*Y1+Z*Z)
25   * +(Z*Z-X2*X2)/2.*ALOG(ABS((Y2+SQRT(X2*X2+Y2*Y2+Z*Z))/(Y1+SQRT(X2*
26   * X2+Y1*Y1+Z*Z)))))-(Z*Z-X1*X1)/2.*ALOG(ABS((Y2+SQRT(X1*X1+Y2*Y2+Z*Z
27   * ))/(Y1+SQRT(X1*X1+Y1*Y1+Z*Z))))-X1*Y2*ALOG(ABS((X2+SQRT(X2*X2+Y2*
28   * Y2+Z*Z))/(X1+SQRT(X1*X1+Y2*Y2+Z*Z))))+X1*Y1*ALOG(ABS((X2+SQRT(X2*
29   * X2+Y1*Y1+Z*Z))/(X1+SQRT(X1*X1+Y1*Y1+Z*Z))))-XD*X1*ALOG(ABS((Y2+
30   * SQRT(X1*X1+Y2*Y2+Z*Z))/(Y1+SQRT(X1*X1+Y1*Y1+Z*Z))))-XD*Y2*ALOG(
31   * ABS((X1+SQRT(X1*X1+Y2*Y2+Z*Z))/(X2+SQRT(X2*X2+Y2*Y2+Z*Z))))+XD*Y1
32   * *ALOG(ABS((X1+SQRT(X1*X1+Y1*Y1+Z*Z))/(X2+SQRT(X2*X2+Y1*Y1+Z*Z))))
33    DPDX(X1,X2,Y1,Y2,Z)=Y2*ALOG(ABS((X1+SQRT(X1*X1+Y2*Y2+Z*Z))/(X2+
34   * SQRT(X2*X2+Y2*Y2+Z*Z))))+Y1*ALOG(ABS((X2+SQRT(X2*X2+Y1*Y1+Z*Z))/
35   * (X1+SQRT(X1*X1+Y1*Y1+Z*Z))))+X1*ALOG(ABS((Y1+SQRT(X2*X2+Y1*Y1+Z*Z
36   * ))/(Y2+SQRT(X2*X2+Y2*Y2+Z*Z))*(Y2+SQRT(X1*X1+Y2*Y2+Z*Z))/(Y1+SQRT
37   * (X1*X1+Y1*Y1+Z*Z))))+XD*ALOG(ABS((Y1+SQRT(X1*X1+Y1*Y1+Z*Z))/(Y2+
38   * SQRT(X1*X1+Y2*Y2+Z*Z))*(Y2+SQRT(X2*X2+Y2*Y2+Z*Z))/(Y1+SQRT(X2*X2+
39   * Y1*Y1+Z*Z))))
40   * +ABS(Z)*(ATAN(X2*Y2/ABS(Z)/SQRT(X2*X2+Y2*Y2+Z*Z))-ATAN(X1*Y2/ABS(
41   * Z)/SQRT(X1*X1+Y2*Y2+Z*Z))-ATAN(X2*Y1/ABS(Z)/SQRT(X2*X2+Y1*Y1+Z*Z)
42   * )+ATAN(X1*Y1/ABS(Z)/SQRT(X1*X1+Y1*Y1+Z*Z)))
```

```
      DSDX(X1,X2,Y1,Y2,Z)=Y2*ALOG(ABS((X1+SQRT(X1*X1+Y2*Y2+Z*Z))/(X2+
     * SQRT(X2*X2+Y2*Y2+Z*Z))))+Y1*ALOG(ABS((X2+SQRT(X2*X2+Y1*Y1+Z*Z))/
     * (X1+SQRT(X1*X1+Y1*Y1+Z*Z))))+X1*ALOG(ABS((Y1+SQRT(X2*X2+Y1*Y1+Z*Z
     * ))/(Y2+SQRT(X2*X2+Y2*Y2+Z*Z))*(Y2+SQRT(X1*X1+Y2*Y2+Z*Z))/(Y1+SQRT
     * (X1*X1+Y1*Y1+Z*Z))))+XD*ALOG(ABS((Y1+SQRT(XL*XL+Y1*Y1+Z*Z))/(Y2+
     * SQRT(XL*XL+Y2*Y2+Z*Z))*(Y2+SQRT(X2*X2+Y2*Y2+Z*Z))/(Y1+SQRT(X2*X2+
     * Y1*Y1+Z*Z))))
      DPDY(X1,X2,Y1,Y2,Z)=SQRT(X2*X2+Y2*Y2+Z*Z)-SQRT(X1*X1+Y2*Y2+Z*Z)
     * -SQRT(X2*X2+Y1*Y1+Z*Z)+SQRT(X1*X1+Y1*Y1+Z*Z)+X1*ALOG(ABS((X1+SQRT
     * (X1*X1+Y2*Y2+Z*Z))/(X2+SQRT(X2*X2+Y2*Y2+Z*Z))*(X2+SQRT(X2*X2+Y1*
     * Y1+Z*Z))/(X1+SQRT(X1*X1+Y1*Y1+Z*Z))))+XD*ALOG(ABS((X2+SQRT(X2*X2+
     * Y2*Y2+Z*Z))/(XL+SQRT(XL*XL+Y2*Y2+Z*Z))*(XL+SQRT(XL*XL+Y1*Y1+Z*Z))
     * /(X2+SQRT(X2*X2+Y1*Y1+Z*Z))))
      D2PDXDY(X1,X2,Y1,Y2,Z)=ALOG(ABS((X1+SQRT(X1*X1+Y2*Y2+Z*Z))/(X2+
     * SQRT(X2*X2+Y2*Y2+Z*Z))*(X2+SQRT(X2*X2+Y1*Y1+Z*Z))/(X1+SQRT(X1*X1+
     * Y1*Y1+Z*Z))))+XD/SQRT(XL*XL+Y1*Y1+Z*Z)-XD/SQRT(XL*XL+Y2*Y2+Z*Z)
      DPDZ(X1,X2,Y1,Y2,Z)=Z*ALOG(ABS((Y2+SQRT(X2*X2+Y2*Y2+Z*Z))/(Y1+SQRT
     * (X2*X2+Y1*Y1+Z*Z))*(Y1+SQRT(X1*X1+Y1*Y1+Z*Z))/(Y2+SQRT(X1*X1+Y2*
     * Y2+Z*Z))))+X1*SGN(Z)*(ATAN(X2*Y2/ABS(Z)/SQRT(X2*X2+Y2*Y2+Z*Z))-
     * ATAN(X1*Y2/ABS(Z)/SQRT(X1*X1+Y2*Y2+Z*Z))+ATAN(X1*Y1/ABS(Z)/SQRT(
     * X1*X1+Y1*Y1+Z*Z))-ATAN(X2*Y1/ABS(Z)/SQRT(X2*X2+Y1*Y1+Z*Z)))+XD*
     * SGN(Z)*(ATAN(XL*Y2/ABS(Z)/SQRT(XL*XL+Y2*Y2+Z*Z))-ATAN(X2*Y2/ABS(Z
     * )/SQRT(X2*X2+Y2*Y2+Z*Z))+ATAN(X2*Y1/ABS(Z)/SQRT(X2*X2+Y1*Y1+Z*Z))
     * -ATAN(XL*Y1/ABS(Z)/SQRT(XL*XL+Y1*Y1+Z*Z)))
      D2PDXDZ(X1,X2,Y1,Y2,Z)=SGN(Z)*(ATAN(X2*Y2/ABS(Z)/SQRT(X2*X2+Y2*Y2+
     * Z*Z))-ATAN(X1*Y2/ABS(Z)/SQRT(X1*X1+Y2*Y2+Z*Z))+ATAN(X1*Y1/ABS(Z)/
     * SQRT(X1*X1+Y1*Y1+Z*Z))-ATAN(X2*Y1/ABS(Z)/SQRT(X2*X2+Y1*Y1+Z*Z)))
     * +XD*Z/(XL*XL+Z*Z)*(Y2/SQRT(XL*XL+Y2*Y2+Z*Z)-Y1/SQRT(XL*XL+Y1*Y1+
     * Z*Z))
      RO(X)=ROO+DRO*(X-XI1(1))
      EL(X)=ALOG(1.0/SIN(RO(X)/2.*PI))/4./PI
      DL(X)=-DRO*COS(RO(X)*PI/2.)/SIN(RO(X)*PI/2.)/8.
      PI=3.1415926
```

THIS PART OF THE PROGRAM DEFINES THE TUNNEL GEOMETRY. HERE IT IS SET UP FOR A SQUARE TUNNEL OF UNIT WIDTH AND HEIGHT. EACH WALL IS DIVIDED INTO FOUR STRIPS AND EACH STRIP IS DIVIDED INTO SIXTEEN ELEMENTS WHICH EXTEND FROM X=-2.6 TO X=1.8.

```
      DO 1 I=1,256,16
      XI1(I)=-2.6
      XI1(I+1)=-2.2
      XI1(I+2)=-1.8
      XI1(I+3)=-1.4
      XI1(I+4)=-1.0
      XI1(I+5)=-.7
      XI1(I+6)=-.45
      XI1(I+7)=-.25
      XI1(I+8)=-.1
      XI1(I+9)=0.0
      XI1(I+10)=.1
      XI1(I+11)=.25
      XI1(I+12)=.45
      XI1(I+13)=.7
      XI1(I+14)=1.0
      XI1(I+15)=1.4
    1 CONTINUE
```

```
      DO 2 I=1,256
      XI2(I)=XI1(I+1)
    2 CONTINUE
      DO 3 I=16,256,16
      XI2(I)=1.8
    3 CONTINUE
      DO 4 I=1,64
      ETA1(I)=.5
      ETA2(I)=.5
      ETA1(I+64)=-.5
      ETA2(I+64)=-.5
      ZETA1(I+128)=.5
      ZETA2(I+128)=.5
      ZETA1(I+192)=-.5
      ZETA2(I+192)=-.5
    4 CONTINUE
      DO 5 I=1,16
      ZETA1(I)=.25
      ZETA1(I+16)=0.0
      ZETA1(I+32)=-.25
      ZETA1(I+48)=-.5
    5 CONTINUE
      DO 6 I=1,64
      ZETA1(I+64)=ZETA1(I)
      ETA1(I+128)=ZETA1(I)
      ETA1(I+192)=ZETA1(I)
    6 CONTINUE
      DO 7 I=1,128
      ZETA2(I)=ZETA1(I)+.25
      ETA2(I+128)=ETA1(I+128)+.25
    7 CONTINUE
      DO 8 I=1,256
      XI(I)=(XI1(I)+XI2(I))/2.
      ETA(I)=(ETA1(I)+ETA2(I))/2.
      ZETA(I)=(ZETA1(I)+ZETA2(I))/2.
    8 CONTINUE
      XTL=10.0
      PRINT 991
```

THIS PART OF THE PROGRAM DEFINES THE WALL CHARACTERISTICS. IN THIS CASE THE SIDE WALLS ARE SOLID AND THE TOP AND BOTTOM WALLS EACH HAVE FOUR CONSTANT WIDTH SLOTS. THE OPEN RATIO OF THE SLOTTED WALLS IS 6 PERCENT.

```
      ROO=.06
      DPO=0.0
      DO 11 I=1,128
      C1(I)=0.0
      C2(I)=0.0
      C3(I)=1.0
      C4(I)=0.0
   11 CONTINUE
      DO 12 I=129,256
      C1(I)=0.0
      C2(I)=1.0
      C3(I)=0.0
      C4(I)=EL(XI(I))
   12 CONTINUE
```

THIS PART OF THE PROGRAM COMPUTES THE INFLUENCE COEFFICIENTS, A(I,J)

```
      DO 190 I=1,256
      NWI=1
      IF(I.GT.64)NWI=2
      IF(I.GT.128)NWI=3
      IF(I.GT.192)NWI=4
      DO 180 J=1,256
      NWJ=1
      IF(J.GT.64)NWJ=2
      IF(J.GT.128)NWJ=3
      IF(J.GT.192)NWJ=4
      X1=XI(I)-XI1(J)
      X2=XI(I)-XI2(J)
      Y1=ETA(I)-ETA1(J)
      Y2=ETA(I)-ETA2(J)
      Z=ZETA(I)-ZETA(J)
      XD=XI2(J)-XI1(J)
      XL=XI(I)-XIL
      IF(NWJ.LT.3)Y1=ZETA(I)-ZETA1(J)
      IF(NWJ.LT.3)Y2=ZETA(I)-ZETA2(J)
      IF(NWJ.LT.3)Z=ETA(I)-ETA(J)
      IF(NWI.NE.NWJ)GO TO 110
      PHI=S(X1,X2,Y1,Y2,Z)
      U=DSDX(X1,X2,Y1,Y2,Z)
      V=0.0
      DVDX=0.0
      II=I-MOD(I,16)+1
      IJ=J-MOD(J,16)+1
      IF(MOD(I,16).EQ.0)II=I-16+1
      IF(MOD(J,16).EQ.0)IJ=J-16+1
      IF(I.EQ.J)V=-2.0*PI*XD/2.0
      IF(II.EQ.IJ.AND.I.GT.J)V=-2.0*PI*XD
      IF(I.EQ.J)DVDX=-2.*PI
      IF(NWJ.EQ.2)V=-V
      IF(NWJ.EQ.3)W=V
      IF(NWJ.EQ.4)W=-V
      IF(NWJ.EQ.2)DVDX=-DVDX
      IF(NWJ.EQ.3)DWDX=DVDX
      IF(NWJ.EQ.4)DWDX=-DVDX
      GO TO 130
  110 CONTINUE
      PHI=P(X1,X2,Y1,Y2,Z)
      U=DPDX(X1,X2,Y1,Y2,Z)
      V=DPDY(X1,X2,Y1,Y2,Z)
      W=DPDZ(X1,X2,Y1,Y2,Z)
      DVDX=D2PDXDY(X1,X2,Y1,Y2,Z)
      DWDX=D2PDXDZ(X1,X2,Y1,Y2,Z)
      IF(NWI.GT.2)GO TO 120
      T=V
      V=W
      W=T
      DT=DVDX
      DVDX=DWDX
      DWDX=DT
  120 CONTINUE
  130 CONTINUE
      IF(NWI.EQ.1)A(I,J)=C1(I)*PHI+C2(I)*U+C3(I)*V+C4(I)*DVDX
```

```
      IF(NWI.EQ.2)A(I,J)=C1(I)*PHI+C2(I)*U-C3(I)*V-C4(I)*DVDX
      IF(NWI.EQ.3)A(I,J)=C1(I)*PHI+C2(I)*U+C3(I)*W+C4(I)*DWDX
      IF(NWI.EQ.4)A(I,J)=C1(I)*PHI+C2(I)*U-C3(I)*W-C4(I)*DWDX
  180 CONTINUE
  190 CONTINUE

THE FOLLOWING STATEMENT INVERTS THE MATRIX A AND PUTS THE INVERSE IN THE
PLACE OF THE ORIGINAL MATRIX

      CALL MATRIX(10,256,256,0,A,256,DET)

THIS PART OF THE PROGRAM COMPUTES THE DISTURBANCE DUE TO THE MODEL. HERE IT
IS SET UP FOR A NUMBER OF VORTEX DOUBLETS LOCATED IN THE HORIZONTAL CENTER-
PLANE OF THE TUNNEL. THE PROGRAM FIRST READS THE NUMBER OF LIFT ELEMENTS TO
BE USED AND THEN READS THE X AND Y VALUES AND THE WEIGHTING FACTOR FOR EACH
ELEMENT.

      READ 992,L,(XA(I),YA(I),WT(I),I=1,L)
      DO 200 I=1,256
      B(I)=0.0
  200 CONTINUE
      DO 299 K=1,L
      XPP=XA(K)
      YPP=YA(K)
      DO 210 I=1,64
      X=XI(I)-XPP
      Y=ETA(I)-YPP
      Z=ZETA(I)
      PHI=Z/(Y*Y+Z*Z)*(1.0+X/SQRT(X*X+Y*Y+Z*Z))/4.0/PI
      U=Z/4./PI/(X*X+Y*Y+Z*Z)**1.5
      V=-(2.*Y*Z+(2.*X**3*Y*Z+3.*X*Y**3*Z+3.*X*Y*Z**3)/(X*X+Y*Y+Z*Z)**1.
     * 5)/4./PI/(Y*Y+Z*Z)**2
      DVDX=-.75/PI*Y*Z/(X*X+Y*Y+Z*Z)**2.5
      B(I)=B(I)+(C1(I)*PHI+C2(I)*U+C3(I)*V+C4(I)*DVDX)*WT(K)
  210 CONTINUE
      DO 220 I=65,128
      X=XI(I)-XPP
      Y=ETA(I)-YPP
      Z=ZETA(I)
      PHI=Z/(Y*Y+Z*Z)*(1.0+X/SQRT(X*X+Y*Y+Z*Z))/4.0/PI
      U=Z/4./PI/(X*X+Y*Y+Z*Z)**1.5
      V=-(2.*Y*Z+(2.*X**3*Y*Z+3.*X*Y**3*Z+3.*X*Y*Z**3)/(X*X+Y*Y+Z*Z)**1.
     * 5)/4./PI/(Y*Y+Z*Z)**2
      DVDX=-.75/PI*Y*Z/(X*X+Y*Y+Z*Z)**2.5
      B(I)=B(I)+(C1(I)*PHI+C2(I)*U-C3(I)*V-C4(I)*DVDX)*WT(K)
  220 CONTINUE
      DO 230 I=129,192
      X=XI(I)-XPP
      Y=ETA(I)-YPP
      Z=ZETA(I)
      PHI=Z/(Y*Y+Z*Z)*(1.0+X/SQRT(X*X+Y*Y+Z*Z))/4.0/PI
      U=Z/4./PI/(X*X+Y*Y+Z*Z)**1.5
      W=(Y*Y-Z*Z+(X**3*Y*Y+X*Y**4-X**3*Z*Z-X*Y*Y*Z*Z-2.*X*Z**4)/(X*X+Y*Y
     * +Z*Z)**1.5)/4./PI/(Y*Y+Z*Z)**2
      DWDX=(X*X+Y*Y-2.*Z*Z)/(X*X+Y*Y+Z*Z)**2.5/4./PI
```

```
      B(I)=B(I)+(C1(I)*PHI+C2(I)*U+C3(I)*W+C4(I)*DWDX)*WT(K)
  230 CONTINUE
      DO 240 I=193,256
      X=XI(I)-XPP
      Y=ETA(I)-YPP
      Z=ZETA(I)
      PHI=Z/(Y*Y+Z*Z)*(1.0+X/SQRT(X*X+Y*Y+Z*Z))/4.0/PI
      U=Z/4./PI/(X*X+Y*Y+Z*Z)**1.5
      W=(Y*Y-Z*Z+(X**3*Y*Y+X*Y**4-X**3*Z*Z-X*Y*Y*Z*Z-2.*X*Z**4)/(X*X+Y*Y
     * +Z*Z)**1.5)/4./PI/(Y*Y+Z*Z)**2
      DWDX=(X*X+Y*Y-2.*Z*Z)/(X*X+Y*Y+Z*Z)**2.5/4./PI
      B(I)=B(I)+(C1(I)*PHI+C2(I)*U-C3(I)*W-C4(I)*DWDX)*WT(K)
  240 CONTINUE
  299 CONTINUE
```

THIS PART OF THE PROGRAM COMPUTES THE SOURCE STRENGTH SLOPES WHICH SATISFY THE BOUNDARY CONDITIONS.

```
      DO 300 I=1,256
      SIGMA(I)=0.0
  300 CONTINUE
      DO 302 I=1,256
      DO 301 J=1,256
      SIGMA(I)=SIGMA(I)-A(I,J)*B(J)
  301 CONTINUE
  302 CONTINUE
```

THIS PART OF THE PROGRAM COMPUTES THE LIFT INTERFERENCE FACTOR. IT READS THE X, Y, AND Z VALUES AT WHICH THE INTERFERENCE IS TO BE COMPUTED.

```
  400 CONTINUE
      READ 993,XC,YC,ZC
      IF(EOF,5)990,401
  401 CONTINUE
      DELTA1=0.0
      DELTA2=0.0
      DELTA3=0.0
      DELTA4=0.0
      DO 410 J=1,64
      X1=XC-XI1(J)
      X2=XC-XI2(J)
      Y=YC-ETA(J)
      Z1=ZC-ZETA1(J)
      Z2=ZC-ZETA2(J)
      XD=XI2(J)-XI1(J)
      XL=XC-XIL
      W=DPDY(X1,X2,Z1,Z2,Y)
      DELTA1=DELTA1+W*SIGMA(J)/2.
  410 CONTINUE
      DO 420 J=65,128
      X1=XC-XI1(J)
      X2=XC-XI2(J)
      Y=YC-ETA(J)
      Z1=ZC-ZETA1(J)
      Z2=ZC-ZETA2(J)
```

```
      XD=XI2(J)-XI1(J)
      XL=XC-XIL
      W=DPDY(X1,X2,Z1,Z2,Y)
      DELTA2=DELTA2+W*SIGMA(J)/2.
  420 CONTINUE
      DO 430 J=129,192
      X1=XC-XI1(J)
      X2=XC-XI2(J)
      Y1=YC-ETA1(J)
      Y2=YC-ETA2(J)
      Z=ZC-ZETA(J)
      XD=XI2(J)-XI1(J)
      XL=XC-XIL
      W=DPDZ(X1,X2,Y1,Y2,Z)
      DELTA3=DELTA3+W*SIGMA(J)/2.
  430 CONTINUE
      DO 440 J=193,256
      X1=XC-XI1(J)
      X2=XC-XI2(J)
      Y1=YC-ETA1(J)
      Y2=YC-ETA2(J)
      Z=ZC-ZETA(J)
      XD=XI2(J)-XI1(J)
      XL=XC-XIL
      W=DPDZ(X1,X2,Y1,Y2,Z)
      DELTA4=DELTA4+W*SIGMA(J)/2.
  440 CONTINUE
      DELTA=DELTA1+DELTA2+DELTA3+DELTA4
      PRINT 993,XC,YC,ZC,DELTA
      GO TO 400
  990 STOP
  991 FORMAT(39H1         X              Y              Z         DELTA)
  992 FORMAT(I2,/3F10.6)
  993 FORMAT(4F10.6)
      END
```

REFERENCES

1. Keller, James D.; and Wright, Ray H.: A Numerical Method of Calculating the Boundary-Induced Interference in Slotted or Perforated Wind Tunnels of Rectangular Cross Section. NASA TR R-379, 1971.

2. Prandtl, L.: Applications of Modern Hydrodynamics to Aeronautics. NACA Rep. 116, 1921.

3. Goodman, Theodore R.: The Porous Wall Wind Tunnel. Pt. IV – Subsonic Interference Problems in a Circular Tunnel. Rep. No. AD-706-A-2 (Contract AF33(038)-9928), Cornell Aeronaut. Lab., Inc., Aug. 1951.

4. Goethert, Bernhard H.: Transonic Wind Tunnel Testing. AGARDograph No. 49, Pergamon Press, 1961.

5. Davis, Don D., Jr.; and Moore, Dewey: Analytical Study of Blockage- and Lift-Interference Corrections for Slotted Tunnels Obtained by the Substitution of an Equivalent Homogeneous Boundary for the Discrete Slots. NACA RM L53E07b, 1953.

6. Baldwin, Barrett S., Jr.; Turner, John B.; and Knechtel, Earl D.: Wall Interference in Wind Tunnels With Slotted and Porous Boundaries at Subsonic Speeds. NACA TN 3176, 1954.

7. Garner, H. C.; Rogers, E. W. E.; Acum, W. E. A.; and Maskell, E. C.: Subsonic Wind Tunnel Wall Corrections. AGARDograph 109, Oct. 1966.

8. Pearcey, H. H.; Sinnott, C. S.; and Osborne, J.: Some Effects of Wind Tunnel Interference Observed in Tests on Two-Dimensional Aerofoils at High Subsonic and Transonic Speeds. AGARD Rep. 296, Mar. 1959.

NASA-Langley, 1972 —— 1 L-8349